
Indice degli argomenti

§ Programmazione assembly

• Definizione e controllo di sottoprogrammi

• Definizione di funzione

• Chiamate a funzioni

• Convenzioni riguardanti l’uso dei registri

• Funzioni ricorsive
– Stack



2

3© 2002-2003 Alberto Montresor, Riccardo Solmi

Funzioni

§ Una funzione (o routine) è una sequenza di istruzioni che esegue un compito 
definito e che viene usata come se fosse una unità di codice.

• Una funzione è una astrazione presente in (quasi) tutti i linguaggi ad alto livello.

• Assegna un nome ad una operazione complessa (fatta da più istruzioni)
• Mi basta sapere cosa fa una funzione il come non mi interessa

• Maggiore leggibilità del codice.

• Riutilizzazione del codice (anche con altri linguaggi di programmazione)

• L’intero programma assembly viene visto come un insieme di funzioni

• Non è una supportata dall’assembly; quest’ultimo fornisce solo le istruzioni e i  
registri che consentono di realizzare delle funzioni.

• E’ un insieme di convenzioni seguite da (tutti) i programmatori assembly

• Convenzioni sull’uso dei registri

• Convenzioni sull’utilizzo dello stack

• NB. Per convenzioni si intendono regole “software” non imposte 
dall’hardware né dall’assemblatore. Se non le seguite peggio per voi!

• Le convenzioni più usate sono quelle del linguaggio C

Esempio di funzioni

Funzione
chiamante

Funzione
chiamata

Valore di ritorno: $v0

Parametro: $a0

Argomento: str
Esempio:
main: la $a0,str

jal length
sw $v0,result
jr $ra

length: move $t0,$a0
li $t1,0

nextChar: lbu $t2,($t0)
beqz $t2,end
addu $t0,$t0,1
addu $t1,$t1,1
j nextChar

end: move $v0,$t1
jr $ra

Chiamata a funzione

Nome



3

5© 2002-2003 Alberto Montresor, Riccardo Solmi

Ruoli: chiamante e chiamato

§ Essere chiamante/chiamato è una questione di “ruolo”

§ Tutte le funzioni sono chiamate

§ Una funzione può essere:

• Foglia: Se non chiama nessuna altra funzione al suo interno

• Non-foglia: Se è sia chiamante che chiamata

§ Esempi:
main: adder:

lw $a0,x add $v0,$a0,$a1

lw $a1,y jr $ra

jal adder

sw $v0,z
Funzione non-foglia Funzione foglia

Chiamata a funzione e ritorno

§ Chiamata a funzione:

• Per chiamare una funzione si usa l’istruzione
jal function-name (jump and link)

• Memorizza l’indirizzo dell’istruzione che segue jal nel 
registro $ra e salta all’etichetta function-name

§ Ritorno da funzione:

• Per tornare da una funzione si usa l’istruzione
jr $ra (jump register)

• Salta all’indirizzo contenuto nel registro $ra

• All’occorrenza si può anche usare un altro registro

Esecuzione di una chiamata a funzione e ritorno

f

p
jal p

jr $ra

0x00400020

0x00400024 ...

1

$ra ?????????

0x004000242

3

4
5

1. Vengono eseguite le istruzioni della funzione chiamante f che precedono 
la chiamata (jal) a p.

2. Chiamata: l’istruzione jal salva l’indirizzo della prossima istruzione da 
eseguire al ritorno in $ra e salta all’inizio della funzione chiamata.

3. Vengono eseguite tutte le istruzioni di p fino alla jr conclusiva.

4. Ritorno: l’istruzione jr salta alla prima istruzione che segue la chiamata.

5. Vengono eseguite le rimanenti istruzioni della funzione chiamante.

Requisiti per una soluzione generale

f

p
jal p

jr $ra

0x00400020

0x00400024 ...

1

$ra ????????? 0x00400024

2
4

5

1. Il registro $ra non è sufficiente se la funzione chiamata è a sua volta un 
chiamante (vedi figura).

2. Desidero passare dei parametri alla funzione chiamata e ottenere un 
risultato

3. Mi fa comodo definire una convenzione per stabilire quali registri vengono 
preservati dopo una chiamata e quali no

4. La funzione può avere bisogno di variabili locali se non bastano i registri.

q

jr $ra

3

0x00400104

jal q0x00400100

0x00400104 ...

67
?

8

Convenzioni – Registri

§ Registri usati per passare parametri e risultati:

• $a0-$a3 argomenti di una funzione

• E se voglio passare più di 4 parametri?

• $v0-$v1 risultato di una funzione (32/64 bit)

§ Registri preservati tra le chiamate 

• $t0-$t9 e $ra non vengono preservati tra le chiamate

• E se ho bisogno di preservarli?

• $s0-$s7 vengono preservati tra le chiamate

• Come si fanno a preservare se li uso?

§ E se non mi bastano i registri che ci sono per scrivere una 
funzione particolarmente complessa?

Convenzioni – Activation frame

§ Ogni funzione necessita di uno spazio di memoria per

• Memorizzare i valori passati alla funzione come argomenti

• Salvare i registri che una funzione può modificare, ma 
che il chiamante si aspetta che vengano preservati

• Fornire spazio per le variabili locali della procedura
(se non mi bastano i registri)

§ Questo spazio viene chiamato

frame di attivazione della funzione

§ Domanda:

• E’ possibile associare ad ogni funzione uno spazio “fisso” in 
memoria?

Frame di attivazione statico per ogni funzione

f: sw $ra,fFrame

li $a0,5

#li $a1-$a3 ...

li $t0,12

sw $t0,pFrame+8

jal p

lw $ra,fFrame

jr $ra

p: sw $ra,pFrame

lw $t0,pFrame+8

addi $t0,$t0,1

sw $t0,pFrame+4

jal q

lw $t0,pFrame+4

addi $v0,$t0,3

lw $ra,pFrame

jr $ra

.data

fFrame: .space 4  # spazio per $ra

pFrame: .space 12 # spazio per $ra, $t0 e il 5 parametro

qFrame: .space 8  # spazio per $s0 e $s1

q: sw $s0,qFrame

sw $s1,qFrame+4

li $s0,1

move $s1,$s0

li $t0,5

lw $s0,qFrame

lw $s1,qFrame+4

jr $ra

Frame di attivazione e funzioni ricorsive

fact: sw $ra,factFrame

ble $a0,1,base

sw $a0,factFrame+4 # salva n

subu $a0,$a0,1

jal fact         # fact(n-1)

lw $t0,factFrame+4 # carica n

mul $v0,$t0,$v0  # n*(n-1)!

j end

base: li $v0,1         # caso base

end: lw $ra,factFrame

jr $ra

.data

factFrame: .word 8 # spazio per $ra e n

• Un singolo frame x funzione 
non è sufficiente nel caso di 
funzioni ricorsive

• Da qui il nome: “frame di 
attivazione” perché è necessario 
un frame per ogni attivazione
(chiamata) di funzione

• Come faccio ad allocare del 
nuovo spazio in memoria ogni 
volta che chiamo una funzione?

Stack

§ Last-in, first-out (LIFO)

• I frame di attivazione di funzione vengono creati e distrutti 
seguendo un modello last- in, first-out

• E’ quindi possibile memorizzarli in uno stack

§ Operazioni sullo stack:

• push: inserisce un frame nello stack, inserendolo in ultima 
posizione

• pop: rimuove l’ultimo frame dallo stack

• read/write nel frame corrente (ultimo elemento) dello stack

Esempio funzionamento Stack 1

Frame
main()

Stack

stack
grows

f

jal g
...

g

jal h
jal i

h

...

...

...

...
jr $ra

i

...

...

...

...
jr $ra

...

...

jr $ra

...
jr $ra

...

Esempio funzionamento Stack 2

Frame
main()

Frame
f()

Stack

stack
grows

f

jal g
...

1

g

jal h
jal i

h

...

...

...

...
jr $ra

i

...

...

...

...
jr $ra

...

...

jr $ra

...
jr $ra

...

Esempio funzionamento Stack 3

Frame
main()

Frame
f()

Stack

stack
grows

f

jal g
...

1

g

jal h
jal i

h

...

...

Frame
g()...

...
jr $ra

i

...

...

...

...
jr $ra

...

...

jr $ra

...
jr $ra

...

2

Esempio funzionamento Stack 4

Frame
main()

Frame
f()

Stack

stack
grows

f

jal g
...

1

g

jal h
jal i

h

...

...

Frame
g()

Frame
h()

...

...
jr $ra

i

...

...

...

...
jr $ra

...

...

jr $ra

...
jr $ra

...

2

3

Esempio funzionamento Stack 5

Frame
main()

Frame
f()

Stack

stack
grows

f

jal g
...

1

g

jal h
jal i

h

...

...

Frame
g()...

...
jr $ra

i

...

...

...

...
jr $ra

...

...

jr $ra

...
jr $ra

...

2

3

4

Esempio funzionamento Stack 6

Frame
main()

Frame
f()

Stack

stack
grows

f

jal g
...

1

g

jal h
jal i

h

...

...

Frame
g()

Frame
i()

...

...
jr $ra

i

...

...

...

...
jr $ra

...

...

jr $ra

...
jr $ra

...

2

3

54

Esempio funzionamento Stack 7

Frame
main()

Frame
f()

Stack

stack
grows

f

jal g
...

1

g

jal h
jal i

h

...

...

Frame
g()...

...
jr $ra

i

...

...

...

...
jr $ra

...

...

jr $ra

...
jr $ra

...

2

3

54

Esempio funzionamento Stack 8

Frame
main()

Frame
f()

Stack

stack
grows

f

jal g
...

1

g

jal h
jal i

h

...

...

...

...
jr $ra

i

...

...

...

...
jr $ra

...

...

jr $ra

...
jr $ra

...

2

3

54

6

Esempio funzionamento Stack 9

Frame
main()

Stack

stack
grows

f

jal g
...

1

g

jal h
jal i

h

...

...

...

...
jr $ra

i

...

...

...

...
jr $ra

...

...

jr $ra

...
jr $ra

...

2

3

54

6

7

Convenzioni – Stack frame

§ Uno stack frame contiene:
• Registri salvati

• Variabili locali e temporanee

• argomenti per le funzioni chiamate 
dalla funzione (massimo numero 
richiesto; si conta dal quinto)

§ E’ accessibile da:
• Stack pointer $sp

• Lo stack cresce verso gli indirizzi
più bassi della memoria
(Per allocare spazio devo sottrarre)

• NB La dimensione del frame
deve essere un multiplo di 8 byte

Argument 5
Argument 6
Argument ...

Variabili
locali e 

temporanee

Registri 
salvati

Area per 
argomenti
chiamate

$sp

Stack
frame

precedente

Stack
frame

Convenzioni per funzioni non foglia

f: subu $sp,$sp,24

sw $s0,20($sp)

sw $s1,16($sp)

sw $ra,12($sp)

...

sw $t1,0($sp)

sw $t0,8($sp)

jal p

lw $t0,8($sp)

...

lw $ra,12($sp)

lw $s1,16($sp)

lw $s0,20($sp)

addu $sp,$sp,24

jr $ra $sp

($s0)
($s1)
($ra)
($t0)

1. Alloco nello stack uno spazio sufficiente a 
contenere tutti i registri che devo salvare, le 
variabili locali e il numero max di argomenti.

2. Salvo i registri $s0-$s7 che intendo usare e $ra.

a. Se la funzione chiamata richiede più di 4 argomenti 
li metto nello stack. 

b. Se voglio preservare i registri $t0-$t9,$a0-$a3,$v0-
$v1 devo salvarli prima della chiamata a funzione e 
ripristinarli dopo.

3. Ripristino i registri salvati
$ra, $s0-$s7.

4. Libero lo spazio sullo stack
allocato all’inizio.

arg6
arg5

IN
IZ

IO
FI

N
E

PE
R

 O
G

N
I

C
H

IA
M

A
T

A

Stack
frame

Convenzioni per funzioni foglia
p: subu $sp,$sp,16

sw $s0,12($sp)
sw $s1,8($sp)
sw $s2,4($sp)
...
lw $s0,12($sp)
lw $s1,8($sp)
lw $s2,4($sp)
addu $sp,$sp,16
jr $ra

q: ...
jr $ra 

$sp

($s0)

1. Alloco nello stack uno spazio sufficiente a 
contenere tutti i registri che devo salvare, e le 
eventuali variabili locali.

2. Salvo i registri $s0-$s7 che intendo usare.

3. Ripristino i registri salvati
$s0-$s7.

4. Libero lo spazio sullo stack
allocato all’inizio.

NB. Se non uso i registri $s0-$s7 e non ho variabili 
locali non ho bisogno di fare nulla!

($s1)
($s2)

IN
IZ

IO
FI

N
E

Stack
frame

Esempio fattoriale (ricorsivo)

main: subu $sp,$sp,8

sw $ra,4($sp)

lw $a0,n

jal fact  # fact(n)

sw $v0,result

lw $ra,4($sp)

addu $sp,$sp,8

jr $ra

fact: subu $sp,$sp,8

sw $ra,4($sp)

ble $a0,1,base

sw $a0,0($sp)    # salva n

subu $a0,$a0,1

jal fact         # fact(n-1)

lw $t0,0($sp)    # carica n

mul $v0,$t0,$v0  # n*(n-1)!

j end

base: li $v0,1         # caso base

end: lw $ra,4($sp)

addu $sp,$sp,8

jr $ra

Esempio fattoriale (iterativo)

fact: ble $a0,1,base  # $a0 n

move $v0,$a0    # $v0 risultato

loop: subu $a0,$a0,1

mul $v0,$v0,$a0 # n*(n-1)*(n-2)…

bgt $a0,2,loop

jr $ra

base: li $v0,1        # caso base

jr $ra

• Non ho bisogno di usare lo stack

• Il risultato è molto più efficiente sia in spazio che in tempo

• NB I compilatori dei linguaggi ad alto livello sono in genere 
abbastanza intelligenti da fare traduzioni iterative anche di 
funzioni ricorsive.

Esempio fattoriale (precalcolato)

fact: ble $a0,10,base # $a0 n

move $v0,$a0    # $v0 risultato

loop: subu $a0,$a0,1

mul $v0,$v0,$a0 # n*(n-1)*(n-2)…

bgt $a0,2,loop

jr $ra

base: sll $a0,$a0,2   # caso precalcolato

lw $v0,table($a0)

jr $ra

.data

table: .word 1,1,2,6,24,120,720,5040,40320,362880,3628800

• Il risultato occupa più spazio ma calcola in tempo costante i risultati più richiesti

• NB I compilatori dei linguaggi ad alto livello non fanno questo genere di 
ottimizzazioni perché non sanno se ne vale la pena. Le macchine virtuali si.

Convenzioni - Errori comuni 1

§ Questa funzione non rispetta le convenzioni; perché?
function1:

subu $sp,$sp,24

...

jal otherfunction

...

addu $sp,$sp,24 

jr $ra

Convenzioni - Errori comuni 2

§ Questa funzione non rispetta le convenzioni; perché?
function2:

subu $sp,$sp,24
sw $ra,20($sp)
...
li $t0,4
jal otherfunction
add $t0,$t0,$v0
...
lw $ra,20($sp)
addu $sp,$sp,24
jr $ra

Convenzioni - Errori comuni 3

§ Questa funzione non rispetta le convenzioni; perché?
function3:

subu $sp,$sp,24
sw $ra,20($sp)
...
li $s0,4
jal otherfunction
add $s0,$s0,$v0
...
lw $ra,20($sp)
addu $sp,$sp,24
jr $ra

Convenzioni - Errori comuni 4

§ Questa funzione rispetta le convenzioni?
function4:

subu $sp,$sp,24
sw $ra,20($sp)
sw $s0,16($sp)
...
li $s0,4
jal otherfunction
add $s0,$s0,$v0
...
lw $s0,16($sp)
lw $ra,20($sp)
addu $sp,$sp,24
jr $ra

Convenzioni – Errori comuni - Soluzioni

§ Function1

• Function1 non è foglia, quindi dovrebbe: 
• salvare il valore del registro $ra all’inizio di Function1

• ricaricarlo alla fine

§ Function2

• La funzione utilizza il registro $t0 prima e dopo della 
chiamata della funzione otherfunction ; 

• $t0 andrebbe salvato prima della chiamata di otherfunction e 
ripristinato subito dopo

§ Function3

• La funzione utilizza il registro $s0 nella funzione;

• $s0 andrebbe salvato all’inizio di function3 e ripristinato 
alla fine

Esercitazione

§ Scrivere una funzione assembly che calcola la funzione di 
Fibonacci per un valore n

• fib(0) = 1;

• fib(1) = 1;

• fib(n) = fib(n-1) + fib(n-2)

§ Provare a scrivere:
• una versione iterativa

• una versione ricorsiva

§ Aggiungere una funzione main() che:

• prenda in input un valore

• chiama fib

• stampa il valore di ritorno

Esercitazione: modalità di consegna

§ Data di consegna: entro il 9/12/2002

§ Metodo di consegna:

• Tramite mail a solmi@cs.unibo.it

• Subject del mail: ARCH-ES02 (siate precisi!)

• Un file assembly con estensione .s in attachment

• Il vostro nome cognome ed eventuali commenti nel corpo 
del messaggio

§ Nota:

• Gli esercizi non sono obbligatori…

• …però sono fortemente consigliati!

• Ne discuteremo in aula o a ricevimento su richiesta


